5. ZADATAK

In[1]:=
```
trans = AffineTransform[{{{2, -1}, {1, 1}}, {5, 7}}]
```

Out[1]=
$$\text{TransformationFunction}\left[\left(\begin{array}{cc|c} 2 & -1 & 5 \\ 1 & 1 & 7 \\ \hline 0 & 0 & 1 \end{array}\right)\right]$$

In[2]:=
```
invtrans = InverseFunction[trans]
```

Out[2]=
$$\text{TransformationFunction}\left[\left(\begin{array}{cc|c} \frac{1}{3} & \frac{1}{3} & -4 \\ -\frac{1}{3} & \frac{2}{3} & -3 \\ \hline 0 & 0 & 1 \end{array}\right)\right]$$

In[3]:=
```
ispisiFle[fle_] := Print["x' = ", fle[[1]], "\ny' = ", fle[[2]]];
```

In[4]:=
```
ispisiFle[invtrans[{x, y}]]
```

$$x' = -4 + \frac{x}{3} + \frac{y}{3}$$
$$y' = -3 - \frac{x}{3} + \frac{2\,y}{3}$$

6. ZADATAK

Funkcija t[O1_, A1_, B1_] određuje matricu afinog preslikavanja koje kanonski trougao OAB preslikava u trougao O1A1B1.

In[5]:=
```
t[O1_, A1_, B1_] := AffineTransform[{Transpose[{A1 - O1, B1 - O1}], O1}];
```

In[6]:=
```
t[{5, -4}, {7, -8}, {4, 1}]
```

Out[6]=
$$\text{TransformationFunction}\left[\left(\begin{array}{cc|c} 2 & -1 & 5 \\ -4 & 5 & -4 \\ \hline 0 & 0 & 1 \end{array}\right)\right]$$

In[7]:=
```
ispisiFle[%[{x, y}]]
```

$$x' = 5 + 2\,x - y$$
$$y' = -4 - 4\,x + 5\,y$$

In[8]:=
```
Det[{{2, -1}, {-4, 5}}]
```

Out[8]= 6

6 > 0 → preslikavanje čuva orjentaciju

In[9]:=
```
povrsina = 6 * 1/2
```

Out[9]= 3

Površina trougla O'A'B' je 3.

7. ZADATAK

O (0, 0), A (1, 0), B (0, 1)

f : O, A, B → P, Q , R

g : O, A , B → P', Q', R'

→ preslikavanje $g \circ f^{-1}$: P, Q , R→ P', Q', R'

In[10]:=
```
Show[Graphics[{
    {Red, Triangle[{{0, 0}, {1, 0}, {0, 1}}]},
    {Blue, Triangle[{{1, 1}, {1, 2}, {4, 4}}]},
    {Green, Triangle[{{5, -4}, {7, -8}, {4, 1}}]}
  }], Axes → True]
```

Out[10]=



In[11]:=
```
f = t[{1, 1}, {1, 2}, {4, 4}]
```

Out[11]=
$$\text{TransformationFunction}\left[\left(\begin{array}{cc|c} 0 & 3 & 1 \\ 1 & 3 & 1 \\ \hline 0 & 0 & 1 \end{array}\right)\right]$$

In[12]:=
```
invf = InverseFunction[f]
```

Out[12]=
$$\text{TransformationFunction}\left[\left(\begin{array}{cc|c} -1 & 1 & 0 \\ \frac{1}{3} & 0 & -\frac{1}{3} \\ \hline 0 & 0 & 1 \end{array}\right)\right]$$

In[13]:=
```
g = t[{5, -4}, {7, -8}, {4, 1}]
```

Out[13]=
$$\text{TransformationFunction}\left[\left(\begin{array}{cc|c} 2 & -1 & 5 \\ -4 & 5 & -4 \\ \hline 0 & 0 & 1 \end{array}\right)\right]$$

In[14]:=
```
Composition[g, invf]
```

Out[14]=
$$\text{TransformationFunction}\left[\left(\begin{array}{cc|c} -\frac{7}{3} & 2 & \frac{16}{3} \\ \frac{17}{3} & -4 & -\frac{17}{3} \\ \hline 0 & 0 & 1 \end{array}\right)\right]$$

In[15]:= **ispisiFle[%[{x, y}]]**

$$x' = \frac{16}{3} - \frac{7\,x}{3} + 2\,y$$

$$y' = -\frac{17}{3} + \frac{17\,x}{3} - 4\,y$$

$g \circ f^{-1}$ dato formulama :

$$x' = \frac{16}{3} - \frac{7\,x}{3} + 2\,y$$

$$y' = -\frac{17}{3} + \frac{17\,x}{3} - 4\,y$$

slika trougao PQR u trougao P ' Q ' R '.

In[16]:= **Det[{{$\frac{-7}{3}$, 2}, {$\frac{17}{3}$, -4}}]**

Out[16]= -2

-2 < 0 → preslikavanje ne čuva orjentaciju

8. ZADATAK

In[17]:= **tp = {0, 0};**

In[18]:= **tq = {5, 5};**

In[19]:= **tr = {10, -15};**

In[20]:= **tp1 = {1, -7};**

In[21]:= **tq1 = {0, 0};**

In[22]:= **tr1 = {19, -8};**

In[23]:= **pq = tp - tq;**

In[24]:= **pr = tr - tp;**

In[25]:= **qr = tr - tq;**

In[26]:= **p1q1 = tp1 - tq1;**

In[27]:= **p1r1 = tr1 - tp1;**

In[28]:= **q1r1 = tr1 - tq1;**

In[29]:= **Norm[pq]**

Out[29]= $5\sqrt{2}$

In[30]:= **Norm[pr]**

Out[30]= $5\sqrt{13}$

In[31]:= **Norm[qr]**

Out[31]= $5\sqrt{17}$

In[32]:=
```
Norm[p1q1]
```

Out[32]= $5\sqrt{2}$

In[33]:=
```
Norm[p1r1]
```

Out[33]= $5\sqrt{13}$

In[34]:=
```
Norm[q1r1]
```

Out[34]= $5\sqrt{17}$

→ Na osnovu stava SSS trougao PQR i trougao P' Q' R' su podudarni

f : O, A, B → P, Q, R

g : O, A, B → P', Q', R'

→ $g \circ f^{-1}$: P,Q,R → P', Q', R'

In[35]:=
```
f = t[tp, tq, tr]
```

Out[35]= TransformationFunction$\left[\left(\begin{array}{cc|c} 5 & 10 & 0 \\ 5 & -15 & 0 \\ \hline 0 & 0 & 1 \end{array}\right)\right]$

In[36]:=
```
invf = InverseFunction[f]
```

Out[36]= TransformationFunction$\left[\left(\begin{array}{cc|c} \frac{3}{25} & \frac{2}{25} & 0 \\ \frac{1}{25} & -\frac{1}{25} & 0 \\ \hline 0 & 0 & 1 \end{array}\right)\right]$

In[37]:=
```
g = t[tp1, tq1, tr1]
```

Out[37]= TransformationFunction$\left[\left(\begin{array}{cc|c} -1 & 18 & 1 \\ 7 & -1 & -7 \\ \hline 0 & 0 & 1 \end{array}\right)\right]$

In[38]:=
```
Composition[g, invf]
```

Out[38]= TransformationFunction$\left[\left(\begin{array}{cc|c} \frac{3}{5} & -\frac{4}{5} & 1 \\ \frac{4}{5} & \frac{3}{5} & -7 \\ \hline 0 & 0 & 1 \end{array}\right)\right]$

In[39]:=
```
ispisiFle[%[{x, y}]]
```

$x' = 1 + \frac{3x}{5} - \frac{4y}{5}$

$y' = -7 + \frac{4x}{5} + \frac{3y}{5}$

In[40]:=
```
ϕ = ArcCos[3/5] // N
```

Out[40]= 0.927295

→ Rotacija za ugao $\phi = \arccos(\frac{3}{5})$ i translacija za vektor v = (1, -7).

Definišemo funcije :

In[41]:=
```
translacija[v_] := {{1, 0, v[[1]]}, {0, 1, v[[2]]}, {0, 0, 1}};
(*translacija za vektor v*)
```

In[42]:=
```
rot0[ϕ_] := {{Cos[ϕ], -Sin[ϕ], 0}, {Sin[ϕ], Cos[ϕ], 0}, {0, 0, 1}};
(*rotacija oko koordinatnog početka za ugao ϕ*)
```

```
In[43]:= rotacija[S_, ϕ_] := translacija[S].rot0[ϕ].translacija[-S];
         (*rotacija oko tačke S za ugao ϕ*)

In[44]:= skal0[λ1_, λ2_] := DiagonalMatrix[{λ1, λ2, 1}];
         (*skaliranje sa centrom u koordinatnom početku i koeficijentima λ1 u λ2*)

In[45]:= skaliranje[S_, λ1_, λ2_] := translacija[S].skal0[λ1, λ2].translacija[-S];
         (*skaliranje sa centrom u tački S i koeficijentima λ1 u λ2*)

In[46]:= homotetija[S_, λ_] := skaliranje[S, λ, λ];
         (*homotetija sa centrom u S i koeficijentom λ - spec. slucaj skaliranja za λ1=λ2=λ*)

In[47]:= ref0[n_] :=
           IdentityMatrix[3] - 2 Outer[Times, Normalize[Append[n, 0]], Normalize[Append[n, 0]]];
         (*refleksija u odnosu na pravu sa vektorom normale n
          koja sadrzi koordinatni pocetak*)

In[48]:= refleksija[P_, n_] := translacija[P].ref0[n].translacija[-P];
         (*refleksija u odnosu na pravu sa vektorom normale n koja sadrzi tacku P*)
```

9. ZADATAK

```
In[49]:= tC = {1, 2};

In[50]:= λ = 3;
```

1. način - koristimo naše funkcije

```
In[51]:= matrica = homotetija[tC, λ]

Out[51]= {{3, 0, -2}, {0, 3, -4}, {0, 0, 1}}

In[52]:= matrica // MatrixForm

Out[52]//MatrixForm=
```

$$\begin{pmatrix} 3 & 0 & -2 \\ 0 & 3 & -4 \\ 0 & 0 & 1 \end{pmatrix}$$

```
In[53]:= matrica.{0, 0, 1}

Out[53]= {-2, -4, 1}
```

2. način - koristimo ugrađene funkcije

```
In[54]:= trans = ScalingTransform[{3, 3}, tC]

Out[54]= TransformationFunction[( 3  0 | -2 )
                                ( 0  3 | -4 )
                                ( 0  0 |  1 )]

In[55]:= trans[{0, 0}]

Out[55]= {-2, -4}

In[56]:= ispisiFle[trans[{x, y}]]

x' = -2 + 3 x
y' = -4 + 3 y
```

→ Koordinatni početak se preslikava u tačku (-2,-4).

In[57]:=
```
Manipulate[Show[Graphics[GeometricTransformation[
    {PointSize[Large], Point[{0, 0}]}, ScalingTransform[{λ, λ}, tC]]],
  Axes → True, PlotRange → {{-2, 2}, {3, -5}}], {λ, 1, 3, Appearance → "Labeled"} ]
```

Out[57]=



10. ZADATAK

In[58]:=
$$\phi = \frac{7\pi}{6};$$

In[59]:=
```
tA = {-2, 3};
```

1. način - koristimo naše funkcije

In[60]:=
```
matrica = rotacija[tA, ϕ]
```

Out[60]=
$$\left\{\left\{-\frac{\sqrt{3}}{2}, \frac{1}{2}, -\frac{7}{2}-\sqrt{3}\right\}, \left\{-\frac{1}{2}, -\frac{\sqrt{3}}{2}, 2+\frac{3\sqrt{3}}{2}\right\}, \{0, 0, 1\}\right\}$$

In[61]:=
```
matrica // MatrixForm
```

Out[61]//MatrixForm=
$$\begin{pmatrix} -\frac{\sqrt{3}}{2} & \frac{1}{2} & -\frac{7}{2}-\sqrt{3} \\ -\frac{1}{2} & -\frac{\sqrt{3}}{2} & 2+\frac{3\sqrt{3}}{2} \\ 0 & 0 & 1 \end{pmatrix}$$

In[62]:= `matrica.{1, 3, 1}`

Out[62]= $\left\{-2-\frac{3\sqrt{3}}{2}, \frac{3}{2}, 1\right\}$

2. način - koristimo ugrađene funkcije

In[63]:= `r = RotationTransform[ϕ, tA]`

Out[63]= $\text{TransformationFunction}\left[\left(\begin{array}{cc|c} -\frac{\sqrt{3}}{2} & \frac{1}{2} & \frac{1}{2}\left(-7-2\sqrt{3}\right) \\ -\frac{1}{2} & -\frac{\sqrt{3}}{2} & \frac{1}{2}\left(4+3\sqrt{3}\right) \\ \hline 0 & 0 & 1 \end{array}\right)\right]$

In[64]:= `r[{1, 3}] // Simplify`

Out[64]= $\left\{-2-\frac{3\sqrt{3}}{2}, \frac{3}{2}\right\}$

In[65]:= `ispisiFle[r[{x, y}]]`

$$x' = \frac{1}{2}\left(-7-2\sqrt{3}\right) - \frac{\sqrt{3}\,x}{2} + \frac{y}{2}$$

$$y' = \frac{1}{2}\left(4+3\sqrt{3}\right) - \frac{x}{2} - \frac{\sqrt{3}\,y}{2}$$

→ Tačka (1,3) se preslikava u tačku $\left(-2-\frac{3\sqrt{3}}{2}, \frac{3}{2}\right)$.

In[66]:=
```
Manipulate[Show[Graphics[GeometricTransformation[
    {PointSize[Large], Red, Point[{1, 3}]}, RotationTransform[t, tA]]],
  Axes → True, PlotRange → {{-5, 5}, {0, 7}}], {t, 0, ϕ, Appearance → "Labeled"} ]
```

Out[66]=



12. ZADATAK

In[67]:= `vn = {3, -4};`

In[68]:=
```
tP = {2, 0}; (* tačka sa prave *)
```

1. način - koristimo naše funkcije

In[69]:=
```
matrica = refleksija[tP, vn]
```

Out[69]= $\left\{\left\{\frac{7}{25}, \frac{24}{25}, \frac{36}{25}\right\}, \left\{\frac{24}{25}, -\frac{7}{25}, -\frac{48}{25}\right\}, \{0, 0, 1\}\right\}$

In[70]:=
```
matrica // MatrixForm
```

Out[70]//MatrixForm=

$$\begin{pmatrix} \frac{7}{25} & \frac{24}{25} & \frac{36}{25} \\ \frac{24}{25} & -\frac{7}{25} & -\frac{48}{25} \\ 0 & 0 & 1 \end{pmatrix}$$

2. način - koristimo ugrađene funkcije

In[71]:=
```
ref = ReflectionTransform[vn, tP]
```

Out[71]= $\text{TransformationFunction}\left[\left(\begin{array}{cc|c} \frac{7}{25} & \frac{24}{25} & \frac{36}{25} \\ \frac{24}{25} & -\frac{7}{25} & -\frac{48}{25} \\ \hline 0 & 0 & 1 \end{array}\right)\right]$

In[72]:=
```
ispisiFle[ref[{x, y}]]
```

$$x' = \frac{36}{25} + \frac{7\,x}{25} + \frac{24\,y}{25}$$

$$y' = -\frac{48}{25} + \frac{24\,x}{25} - \frac{7\,y}{25}$$